# A CORRECTION ALGORITHM FOR LOW DISPERSION SWP SPECTRA

## I. Introduction

In this memorandum, we present the algorithm which has been adopted jointly by ESA, NASA, and the SRC for correction of low dispersion SWP spectra that were processed with the bad Intensity Transfer Function (ITF). The algorithm uses a wavelength dependent correction that was derived from the ITF images. This correction is applied to the samples in the spatially-resolved (line-by-line) spectra (contained in the 4th and, in the case of exposures in both apertures, in the 7th file of the user's tape). While less accurate than some algorithms tested, this algorithm is generally applicable, introduces errors which are usually less than the intrinsic photometric errors in properly processed spectra, and can be applied with a minimum of programming. (This algorithm may fail in some cases of very weak spectra with negative backgrounds.) The photometric accuracy both of the IUE spectra and of this algorithm is discussed below. The resulting net spectra can be calibrated by the mean sensitivity curve derived by Bohlin and Snijders (1978) with additional errors due to the ITF change being less than 5%.

We include FORTRAN code for a subroutine, SWPFIX, which applies the correction and generates new gross and background spectra.

## II. The Transformation Function

The Flux Number (FN) assigned to an individual pixel can be corrected by a linear equation

$$FN_n = A_i(\lambda)\ FN_o + B_i(\lambda)$$

where $FN_o$ and $FN_n$ represent Flux Numbers assigned by the bad ITF and by the good ITF respectively and the subscript i refers to ranges of $FN_o$ between the various ITF levels. The coefficients for this transformation equation have been derived from "spectra" extracted from the bad ITF image and its replacement. These coefficients for large and small aperture spectra are given in Table 1 and Table 2 respectively. For $FN_o \leq 1084$, the equation is

$$FN_n = 0.960\ FN_o$$

regardless of wavelength.

TABLE 1

COEFFICIENTS FOR CORRECTION OF LARGE APERTURE FLUXES

| λ | $1084 < FN_o \leq 2141$ | | $2141 < FN_o \leq 2684$ | | $2684 < FN_o \leq 4291$ | | L3 |
|---|---|---|---|---|---|---|---|
| | $A_2$ | $B_2$ | $A_3$ | $B_3$ | $A_4$ | $B_4$ | |
| 1100 | 0.2561 | 763 | 1.3674 | -1616 | 1.3670 | -1615 | 2708 |
| 1150 | 0.2517 | 768 | 1.3830 | -1654 | 1.3644 | -1604 | 2705 |
| 1200 | 0.2725 | 746 | 1.3560 | -1574 | 1.3602 | -1585 | 2700 |
| 1250 | 0.2791 | 738 | 1.3907 | -1642 | 1.3441 | -1516 | 2681 |
| 1300 | 0.2952 | 721 | 1.3771 | -1595 | 1.3383 | -1492 | 2674 |
| 1350 | 0.3103 | 704 | 1.3421 | -1504 | 1.3399 | -1499 | 2676 |
| 1400 | 0.3226 | 691 | 1.3252 | -1455 | 1.3375 | -1488 | 2673 |
| 1450 | 0.3415 | 671 | 1.2900 | -1360 | 1.3366 | -1484 | 2672 |
| 1500 | 0.3415 | 671 | 1.3073 | -1397 | 1.3309 | -1460 | 2665 |
| 1550 | 0.3623 | 648 | 1.2581 | -1270 | 1.3333 | -1470 | 2668 |
| 1600 | 0.3557 | 655 | 1.2860 | -1336 | 1.3284 | -1449 | 2662 |
| 1650 | 0.3576 | 653 | 1.2580 | -1274 | 1.3366 | -1484 | 2672 |
| 1700 | 0.3482 | 664 | 1.2602 | -1289 | 1.3424 | -1509 | 2679 |
| 1750 | 0.3415 | 671 | 1.2662 | -1309 | 1.3449 | -1520 | 2682 |
| 1800 | 0.3321 | 681 | 1.2729 | -1333 | 1.3491 | -1538 | 2687 |
| 1850 | 0.3160 | 698 | 1.2669 | -1337 | 1.3627 | -1596 | 2703 |
| 1900 | 0.2573 | 762 | 1.3603 | -1599 | 1.3688 | -1622 | 2710 |
| 1950 | 0.2271 | 795 | 1.3945 | -1704 | 1.3766 | -1656 | 2719 |
| 2000 | 0.1968 | 828 | 1.4132 | -1776 | 1.3899 | -1713 | 2734 |

TABLE 2

COEFFICIENTS FOR CORRECTION OF SMALL APERTURE FLUXES

| λ | $1084 < FN_o \leq 2141$ | | $2141 < FN_o \leq 2688$ | | $2688 < FN_o \leq 4291$ | | L3 |
|---|---|---|---|---|---|---|---|
| | $A_2$ | $B_2$ | $A_3$ | $B_3$ | $A_4$ | $B_4$ | |
| 1100 | 0.2271 | 795 | 1.4316 | -1784 | 1.3636 | -1600 | 2704 |
| 1150 | 0.2431 | 777 | 1.4191 | -1740 | 1.3576 | -1574 | 2697 |
| 1200 | 0.2621 | 757 | 1.3906 | -1659 | 1.3550 | -1563 | 2694 |
| 1250 | 0.2668 | 752 | 1.4018 | -1678 | 1.3483 | -1534 | 2686 |
| 1300 | 0.2971 | 719 | 1.3407 | -1515 | 1.3491 | -1538 | 2687 |
| 1350 | 0.3065 | 709 | 1.3175 | -1456 | 1.3508 | -1545 | 2689 |
| 1400 | 0.3150 | 699 | 1.3228 | -1458 | 1.3433 | -1513 | 2680 |
| 1450 | 0.3169 | 697 | 1.3094 | -1427 | 1.3466 | -1527 | 2684 |
| 1500 | 0.3340 | 679 | 1.2739 | -1333 | 1.3474 | -1531 | 2685 |
| 1550 | 0.3444 | 668 | 1.2820 | -1340 | 1.3375 | -1488 | 2673 |
| 1600 | 0.3529 | 658 | 1.2603 | -1284 | 1.3391 | -1495 | 2675 |
| 1650 | 0.3576 | 653 | 1.2486 | -1254 | 1.3399 | -1499 | 2676 |
| 1700 | 0.3557 | 655 | 1.2407 | -1239 | 1.3441 | -1516 | 2681 |
| 1750 | 0.3368 | 676 | 1.2661 | -1314 | 1.3483 | -1534 | 2686 |
| 1800 | 0.3245 | 689 | 1.2690 | -1333 | 1.3559 | -1567 | 2695 |
| 1850 | 0.3236 | 690 | 1.2616 | -1318 | 1.3593 | -1582 | 2699 |
| 1900 | 0.2933 | 723 | 1.3143 | -1463 | 1.3610 | -1589 | 2701 |
| 1950 | 0.2687 | 750 | 1.3368 | -1537 | 1.3696 | -1626 | 2711 |
| 2000 | 0.2365 | 785 | 1.3311 | -1559 | 1.3943 | -1732 | 2739 |

In addition to the replacement of the bad ITF image, the SWP ITF was modified by a slight adjustment to the effective exposure times assigned each of the levels. This modifies the photometric correction because the FN of a level is defined by the effective exposure time multiplied by a constant, 61.87 for the SWP. This adjustment causes the 4% change to $FN_o$ below 1084. For $FN_o$ above 4291, the adjustment is less than 1% and is ignored in the transformation presented here.

At 2000 Å, these coefficients are somewhat inaccurate because the extraction slit begins to cross the edge of the camera's target. The 2000 Å coefficients given here were derived from a sample of the spatially-resolved orders that had not reached the edge of the target combined with extrapolated values from other orders. Fluxes from the SWP camera longward of 1950 Å should be regarded as uncertain, at least until further calibration work is completed.

The $FN_o$ at the boundary between the zones where the i=3 and the i=4 coefficients are valid is a function of wavelength. This value of $FN_o$ is given in Tables 1 and 2 as L3. The use of a mean boundary location, as suggested in the heading of the tables, will result in errors of only a few tenths of a percent because there is little change in slope between zones i=3 and i=4. The boundaries between the other zones are constant. In our programs we use constant boundaries for all zones because the use of L3 was found to not improve the correction.

## III. The Correction of Spectra from the Spatially-Resolved File

The spatially-resolved file consists of 55 lines (or orders) parallel to the direction of the dispersion. Each line consists of three arrays of data: the wavelengths, the quality factors, and the flux numbers in the samples extracted along that line. The 28th line is centered on the dispersion line and is assigned an order number of 100. The spatially-resolved file is also called the essr, the line-by-line, and the 4th file.

The sample area in these spectra is twice as large as a pixel. Each sample includes portions of 4 to 7 pixels. To use the transformation formulae given above, it is necessary to know the FN values for individual pixels.

In the present algorithm the intensity level is assumed to be constant in the sample area. The Flux Number per pixel is found by dividing the sample value by 2, its area in pixels. The $FN_o$ is then transformed into $FN_n$ and renormalized to the sample area by multiplication by 2. Finally, new gross and background spectra are extracted from the spatially-resolved file by summing samples at constant wavelength from the appropriate lines. The IUE Spectra Image Processing System (SIPS) extracts the gross spectrum from the equivalent of lines 24 through 32 inclusive for point-source spectra (either aperture) and from lines 21 through 35 for extended-source, large aperture spectra. The small aperture background is extracted from the equivalent of lines 18 through 22 and lines 34 through 38. The large aperture background is extracted from the equivalent of lines 15 through 19 and lines 37 through 41 by IUE SIPS.

The approximation used here is quite good for spectra of extended sources, for trailed spectra, and for background levels. The photometric errors will be larger for samples in which the intensity level varies significantly. Therefore, line profiles and spatial information perpendicular to the dispersion may be less accurate. A detailed discussion of the errors is given below.

In the Appendix, we provide the listing of a subroutine, SWPFIX, which corrects the lines from the spatially-resolved file and extracts a gross and a background spectrum. The subroutine is coded in FORTRAN IV. It is intended to be called after each line of data to be corrected has been read by a main program. We have not attempted to specify the I/O statements for reading the data tape because these are machine dependent.

The input parameters to SWPFIX are:

ILINE, the number of the spectrum line (i.e. order number minus 72) being input to the subroutine for processing.

NDATA, the length of the data arrays FLUX, WAVE, GROSS, and BACK in the calling program. (The arrays are 602 words long on tape.)

LSAP, a flag which is set to 1 for spectra obtained through the large aperture, and to 2 for small aperture spectra.

I1,I2, the spectrum line number of the first and last orders containing the gross spectrum. To duplicate IUE SIPS, these are
I1=24 I2=32 for all point-source and small aperture spectra, and
I1=21 I2=35 for extended-source spectra in the large aperture.

K1,K2, the spectrum line numbers for the first and last orders containing the background at the top of the spectrum. To duplicate IUE SIPS, these are

K1=18 K2=22 for small aperture spectra, and

K1=15 K2=19 for large aperture spectra.

K3,K4, the spectrum line numbers for the first and last orders containing background at the bottom of the spectrum. IUE SIPS uses the equivalent of

K3=34 K4=38 for small aperture spectra, and

K3=37 K4=41 for large aperture spectra.

FLUX, the array containing the $FN_o$ values for samples in the line.

WAVE, the array giving the wavelengths assigned to each sample in the FLUX array.

The output parameters are:

GROSS, the array which contains the sum of the corrected data from lines I1 through I2 inclusive.

BACK, the sum of the corrected data from lines K1 through K2 and from lines K3 through K4. When SWPFIX is called with ILINE=K4+1, the BACK array will be scaled to have an effective area equal to that of GROSS.

SWPFIX must be called with ILINE=0 to initialize the GROSS and BACK arrays prior to the first call with spectrum data.

The arrays FLUX, WAVE, GROSS, and BACK are floating point. The first two words of these arrays are not used as is the standard practice in the merged spectra file (the 5th and, if a double exposure, the 8th file on a user's tape - also called the eslo file).

We summarize rules for usage of SWPFIX here. Before applying it to a spectrum, it should be called with ILINE=0 to initialize the data arrays. It should be called with one new line of data at a time. When called with ILINE=K4+1 the background spectrum will be normalized to have the same area as the gross spectrum. Lines that are not in the gross or background spectra will not be corrected. SWPFIX does not build an array of quality flags for the extracted spectra; if the user is extracting his spectra from different lines than used by IUE SIPS he should merge the quality flags from the appropriate orders.

The output arrays generated by SWPFIX contain information equivalent to the gross and background spectra of the merged spectra file if the user has used the appropriate values of the input parameters. To generate the equivalent of the net spectrum, the background array must be smoothed and subtracted from the gross. At present, IUE SIPS smooths the background by twice passing a 15-point running average filter over it. The 15-point average filter creates a new array in which each entry is the mean of the sample in the corresponding position in the original array with the 7 samples on either side of it. The IUE SIPS filtering does not completely remove discrepant points, such as reseaux, radiation spots, and tube blemishes, but transforms them into broad wiggles in the background. For this reason, IUE SIPS will be changed to include a median filtering (Turnrose, Harvel, and Bohlin 1979) that will eliminate the discrepant points. The median filter creates a new array in which each entry is the median of the 31 samples nearest the corresponding position in the original array. Following the median filtering, IUE SIPS will still apply the 15-point average filter twice. The user is recommended to use as much of this filtering as is practical on his machine. Care is necessary when filtering near the end of the extracted and corrected data. We recommend that, as the end of the data is approached, the number of samples used in the filtering be reduced so that the size of the filtering window just reaches the last valid data point.

Users who wish to avoid smoothing their background can transform the smoothed background produced by IUE SIPS into a flux-corrected background. The smoothed background spectrum produced by IUE SIPS can be re-created by differencing the gross and the net spectra from the merged spectra file (the 5th, 8th, or eslo file). Let $G_i$ represent entries from the gross spectrum, $N_i$ represent entries from the net spectrum, and $R_i$ represent entries in a new array the user wishes to transform into a flux-corrected smoothed background. The initial values for $R_i$ are generated by the equation

$$R_i = \begin{cases} 2*(G_i-N_i)/17 & \text{if a point-source or small aperture spectrum (HT=9 in header).} \\ 2*(G_i-N_i)/29 & \text{if an extended source in large aperture (HT=15 in header).} \end{cases}$$

The flux transformation is accomplished by calling SWPFIX with the R array specified in place of FLUX and with ILINE=99, K3=99, and K4=99. Calling SWPFIX with ILINE=100 and K4=99 will then correctly normalize the BACK array.

After a smoothed background spectrum has been generated, the user can derive the net spectrum by subtracting the background from the gross. This new net spectrum should be equivalent to the net provided by IUE SIPS in the merged spectrum file.

There is a situation where the background spectrum derived from the spatially-resolved files will be wrong. Turnrose and Harvel (1979) report that if the extracted spectra are entirely negative or if the absolute value of the most negative background exceeds the maximum positive flux the data files were incorrectly scaled by IUE SIPS prior to 8 June 1979 at Goddard and prior to 12 July 1979 at VILSPA. They state that this situation can be detected by comparing the background spectrum from the merged spectrum file with the difference between the gross and the net spectra from that file. If there is a scaling error in the merged spectra file, there may also be one in the spatially-resolved file. A user who suspects this error in his spectra should compare his derived background spectrum from the spatially-resolved file with the difference between the gross and the net spectra from the merged spectrum file. If the error exists, he should request reprocessing from the appropriate agency.

The only valid spectral data is on the circular target of the camera. In reading out an image, however, an area larger than the target is scanned. The IUE SIPS spectral extraction programs cross the edge of the target at the long wavelength end of the SWP spectrum so that there are samples near the long wavelength limit which do not contain valid data. The wavelength at which the edge of the target is reached varies with order in the spatially-resolved spectra. In the large aperture spectra, the user should not trust samples at wavelengths longer than (1946+1.10*ILINE) Å. For the small aperture, this limit is approximately (1969+1.12*ILINE) Å. SWPFIX does not check for these limits.

Since only those orders which are used are corrected, SWPFIX should be reasonably rapid. An equivalent program written in FORTH for a PDP 11/40 corrects the entire file in 5.7 minutes. If only the extracted orders are corrected, the execution time should be cut to 2.5 to 3 minutes for point-source and extended-source spectra respectively. We estimate the CPU time on an IBM 360/75 to be on the order of 1 to 2 seconds.

## IV. The Photometric Accuracy of Correctly Processed SWP Spectra

To evaluate the quality of the correction algorithm, it is necessary to know the accuracy of SWP spectra when correctly processed. Photometric accuracy can be considered as two quantities, repeatability and linearity. Repeatability is the accuracy with which a spectrum is reproduced when the same, non-variable source is re-observed under identical conditions. Linearity is the accuracy with which the derived fluxes scale when the same, non-variable source is re-observed under differing conditions, for example, with different exposure times or different background levels.

To investigate these qualities, several series of spectra were obtained of the early-type stars used in the photometric calibration program. These series included spectra obtained with identical exposure times, with varied exposure times, and with the background level increased by added exposures of the tungsten flood lamps. For each series, a well exposed, large aperture spectrum is chosen as the reference spectrum to which the others are compared. The other net spectra (as provided by IUE SIPS) were divided by the reference spectrum on a sample-by-sample basis. The resulting ratios were plotted and averaged in bins of various sizes.

Table 3 summarizes the results of this investigation. The standard star, its spectral type, and the image numbers are identified. The column labeled "LEVEL" gives a percentage exposure time where the exposure time of the reference spectrum is defined to be 100%. Where the average background for the spectrum is significantly different than zero, it is given in FN per pixel. (The background is artificially added by lamp exposures because these stars all have short optimum exposure times.) The mean error, $100*(FN_{test}-FN_{ref})/FN_{ref}$, is given for the 600 Å band between 1250 Å and 1850 Å. This band which excludes Lyman alpha was chosen because slight wavelength misalignments in the extracted spectra could cause large fluctuations in the error at Lyman alpha that have no bearing on the photometric accuracy. The RMS error is given for the same band. The mean error in a 50 Å band centered at 1200 Å is also given. This smaller band includes Lyman alpha and a reseau at 1190 Å.

TABLE 3

REPRODUCIBILITY AND LINEARITY*

| TARGET | IMAGE | LEVEL | BACKGROUND PER PIXEL | TYPE | MEAN ERROR | RMS | ERROR AT 1200 Å | REFERENCE SPECTRUM |
|---|---|---|---|---|---|---|---|---|
| HD 60753 B3 IV | SWP6822 | 100% | - | POINT | -2.2% | 5.0% | -3.2% | SWP6823 |
| | SWP6827 | 100 | - | " | -0.6 | 7.4 | 29.1 | " |
| | SWP6826 | 50 | - | " | -1.8 | 7.2 | -1.3 | " |
| | SWP6824 | 50 | 2118 | " | -5.0 | 10.7 | -3.2 | " |
| | SWP6825 | 50 | 5536 | " | 5.8 | 21.2 | -38.8 | " |
| | SWP3223 | 60 | - | TRAILED | 1.1 | 3.7 | -1.9 | SWP3219 |
| | SWP3220 | 50 | - | " | 0.5 | 3.9 | 0.4 | " |
| | SWP3221 | 40 | - | " | 0.5 | 4.5 | -2.9 | " |
| | SWP3222 | 30 | - | " | -0.2 | 4.4 | -2.8 | " |
| +33°2642 B2 IV | SWP4006 | 50 | - | POINT | -3.1 | 7.3 | -7.3 | SWP4003 |
| | SWP4007 | 30 | - | " | -4.2 | 9.1 | -17.3 | " |
| | SWP4004 | 10 | - | " | -19.6 | 16.0 | -47.8 | " |
| | SWP4005 | 10 | - | " | -15.7 | 28.6 | -49.4 | " |
| +28°4211 sdO | SWP6944 | 100 | - | POINT | 0.4 | 5.6 | 0.2 | SWP6943 |
| | SWP2138 | 65 | 3490 | " | -5.5 | 17.4 | -13.6 | SWP2139 |
| HD 93521 O9 Vp | SWP1956 | 100 | - | POINT | 4.4 | 14.0 | -3.0 | SWP1955 |

* This study uses spectra from images processed with the good ITF.

The data for the 100% spectra indicate that a single spectrum may have systematic errors of at least 4% and that the RMS errors for individual samples are at least 5%, even for spectra taken under the best conditions. For these 4 spectra the average absolute value of the mean error is 1.9% and the average RMS error is 8.5%. The mean error might be improved somewhat if the temperature dependence of the camera sensitivity were included in the calculations. An earlier study of spectra processed with the bad ITF indicated that 3% errors in 100 Å bands could be improved to 2% errors by assuming the sensitivity of camera changed by -0.8% per degree Centigrade.

The accuracy of the spectra taken to investigate linearity is reasonable when the level of repeatibility is considered. It is clear that the accuracy degrades as the exposure level is decreased and as the background level is increased. Such variations, but on a larger scale, were the clue in the past that there was an error in the SWP ITF. However, in this case, the series of trailed spectra with their improved signal to noise demonstrate that the ITF is linear to a few percent in the range tested. Only the very underexposed spectra, SWP4004 and SWP4005, have major errors in the 600 Å bands. Whether these errors are repeatible and what causes them are not yet known.

For the 12 point-source spectra included in this study, the average absolute value of the mean error is 5.7% and the average RMS error is 12.5%.

## V. The Accuracy of the Correction Algorithm

The algorithm was tested by being applied to a sample of standard star spectra and of spectra of representative exotic sources, including emission line sources, extended sources, and very faint sources. The goal of this testing was to show that the application of the algorithm improved the bad spectra without significantly degrading the accuracy of spectra that were handled adequately by the bad ITF. Each image included here was processed twice by IUE SIPS, once with the bad ITF and once with the good ITF. All other variables, such as dispersion constants and geometrical correction files, were kept the same. The spatially-resolved spectra that had been processed with the bad ITF were corrected by SWPFIX and by an equivalent program on the PDP 11/40 mini computer and a net spectrum was extracted from them. For comparison, a similar net spectrum was extracted from the correctly-processed spatially-resolved file. The two net spectra were ratioed, sample by sample, to test the quality of the correction process.

A summary of the results of this testing is given in Table 4. The first two columns are self-explanatory. The column labeled "TYPE" gives "P" for point source, "E" for extended source, and "T" for trailed spectrum. The next two columns give the exposure time in minutes and the background level in FN/pixel. Columns 6 and 7 give an estimate of the maximum net exposure level in the continuum and in emission lines. "S" represents a strong exposure, "M" medium, and "W" weak. "A" represents either very weak or absent and the "X" is given to a spectrum that is known to have some saturation. The mean errors in column 8 are $100*(FN_{o-corr}-FN_{n})/FN_{n}$ averaged from 1150 Å to 1950 Å. The Lyman alpha was included because there should be no wavelength misalignment. The RMS errors are given for the same band.

Figures 1 through 4 show ratios, $100*FN_{n}/FN_{corr}$, for some of the representative spectra from Table 4.

The mean errors for all of these spectra are always less than 3% and usually less than 1%. These are quite low compared with the repeatibility and linearity errors found for correctly processed spectra. The RMS errors also are generally better than the RMS errors given for spectra with comparable background levels in Table 3. (It might be noted that 14 of the spectra included in Table 3 are also in Table 4.) Fig. 1-4 show that, while the mean errors are low, there may still be deviations on a scale of 50 to 100 Å whose amplitude is of the same order as the RMS error. Such fluctuations also appear in the plots generated for the study of reproducibility. Hence, broad and extremely weak features in the user's spectrum, whether processed with the new ITF or corrected by the algorithm, should not be regarded as real.

It should be noted that the spectrum SWP6825 in Table 4 has a background level that places the spectrum entirely above the FN level that is corrected by this algorithm. It has mean and RMS errors that are non-zero because of the small adjustments to the effective exposure times discussed in Section II above. While the changes are small, 1% or less, they will be amplified if a weak spectrum is superimposed on a very high background.

All comparisons described above were done between spectra extracted from the spatially-resolved files. A comparison was made between the net in the merged spectra file and a net spectrum extracted from the spatially-resolved

TABLE 4

THE ACCURACY OF THE CORRECTION ALGORITHM

| TARGET | IMAGE | TYPE | EXP TIME | BKGD/PIXEL | EXPOSURE CONT | EXPOSURE LINES | CORRECTION ERROR MEAN | CORRECTION ERROR RMS |
|---|---|---|---|---|---|---|---|---|
| +28°4211 | SWP2139 | P | 0.43 min | -38 | S | A | -1.5% | 1.8% |
| | SWP2138 | " | 0.28 | 3490 | M | " | 0.7 | 3.7 |
| HD 60753 | SWP4315 | P | 0.17 | -169 | S | A | -0.6 | 2.7 |
| | SWP6823 | " | 0.17 | -102 | S | " | -0.5 | 2.0 |
| | SWP6826 | " | 0.08 | -109 | M | " | -1.2 | 3.0 |
| | SWP6824 | " | 0.08 | 2118 | M | " | 1.4 | 9.0 |
| | SWP6825 | " | 0.08 | 5536 | M | " | -0.8 | 8.0 |
| | SWP3219 | T | 0.68 | – | S | " | -0.7 | 3.8* |
| | SWP3223 | " | 0.41 | – | M | " | -0.2 | 0.8 |
| | SWP3220 | " | 0.34 | – | M | " | -0.6 | 1.0 |
| | SWP3221 | " | 0.27 | – | M | " | -0.5 | 1.0 |
| | SWP3222 | " | 0.20 | – | M | " | -0.2 | 1.3 |
| +33°2642 | SWP4003 | P | 4.0 | -30 | S | A | -0.8 | 2.0 |
| | SWP4006 | " | 2.0 | 0 | M | " | 0.3 | 2.7 |
| | SWP4007 | " | 1.2 | 33 | M | " | -0.6 | 3.2 |
| | SWP4004 | " | 0.4 | – | W | " | -0.6 | 23.3 |
| | SWP4005 | " | 0.4 | – | W | " | -1.3 | 17.8 |
| ζ Cas | SWP4316 | P | 0.02 | 71 | X | A | -0.4 | 1.5 |
| η UMa | SWP2341 | T | 0.005 | – | S | A | -0.3 | 0.8 |
| Capella | SWP4626 | P | 0.67 | -62 | S | S | 0.1 | 3.8 |
| Nova Cyg | SWP3886 | P | 3 | – | A | S | 0.1 | 32.4 |
| SMC X-1 | SWP6219 | P | 37 | 90 | S | A | -0.6 | 2.0 |
| CI Cyg | SWP5485 | P | 45 | 110 | W | S | 0.4 | 2.7 |
| U Sco | SWP5728 | P | 40 | 108 | M | S | 0.1 | 2.8 |
| LMC X-4 | SWP6220 | P | 45 | 152 | S | A | -0.1 | 1.9 |
| V603 Aql | SWP5921 | P | 15 | 160 | W | M | -1.2 | 1.9 |
| HR Del | SWP5918 | P | 20 | 180 | S | M | -1.3 | 2.1 |
| NGC 6093 | SWP6026 | E | 95 | 481 | M | A | 2.0 | 3.6 |
| RU Lup | SWP5569 | P | 180 | 985 | M | M | -2.7 | 3.9 |
| N 63A | SWP3490 | E | 395 | 2800 | S | M | 0.0 | 0.6 |
| N 49 | SWP2115 | E | 385 | 2840 | W | M | -0.8 | 6.2 |
| 0837-120 | SWP4292 | P | 392 | 3200 | W | W | -1.6 | 6.4 |

*Dispersion constant error.

file for a correctly processed image. This comparison showed mean errors in 50 Å bins of 0 to 2%. The RMS error was about 1.5%. We suspect that these errors originate in the extraction process. We have not studied enough spectra to known which net, if either, has better photometric accuracy.

A second test of the correction algorithm was made with spectra of emission line sources. This test compared emission line fluxes derived from spectra processed with the good ITF with fluxes from the same spectra processed with the bad ITF and then corrected. Table 5 summarizes the results of this test when carried out on the spectra of Capella, Nova Cygni 1978, and CI Cygni. The errors that would be made in these fluxes if the bad ITF spectra were used are given as well as the errors in the fluxes after the correction algorithm is used. The errors after correction are given relative to the net spectrum from the merged spectra file as well as the net derived from the spatially-resolved file. Relative to the spatially-resolved file, the errors average 2.1%, but relative to the merged spectrum they are more than twice as large, reaching 10% in some cases. At present, we do not have an explanation for this error. It could be related to differences between these two files mentioned above or it could be software errors in one of our extraction programs.

## VI. The Absolute Calibration

At the present there is no evidence that the mean sensitivity curve (Bohlin and Snijders, 1978; Bohlin et al. 1980) will require significant changes. A new sensitivity curve derived from a small sample of spectra processed with the new ITF shows deviations from the old curve of less than 5%. The absolute sensitivity curve will be discussed in more detail elsewhere.

## VII. Documentation

To avoid ambiguity in published results, we request that users who have used this algorithm to correct their data state this in their papers. We suggest the following phrase:
"SWP spectra processed with the incorrect Intensity Transfer Function were corrected with the 3 Agency 4th file method."

A. Cassatella
A. Holm
D. Ponz
F.H. Schiffer, III

TABLE 5

THE ACCURACY OF CORRECTED EMISSION LINE FLUXES

| TARGET | IMAGE | λ | NET FN | CORRECTION ERROR RELATIVE TO SPATIALLY-RESOLVED | MERGED | ERROR IN UNCORRECTED SPECTRUM |
|---|---|---|---|---|---|---|
| Capella | SWP4626 | 1216 Å | 147010 | -1.7% | +0.4% | +7.1% |
| | | 1240 | 4980 | -0.5 | +5.2 | +4.6 |
| | | 1306 | 28850 | -0.7 | -9.7 | +15.5 |
| | | 1335 | 20760 | -4.2 | -3.6 | +21.1 |
| | | 1355 | 6960 | -0.7 | -9.2 | +4.2 |
| | | 1550 | 18210 | +5.1 | +1.7 | +19.7 |
| Nova Cyg 1978 | SWP3886 | 1240 | 84420 | -2.7 | -4.3 | +14.1 |
| | | 1335 | 15520 | +2.3 | +6.7 | +7.4 |
| | | 1400 | 20710 | +2.0 | +5.6 | +8.4 |
| | | 1486 | 72330 | -1.6 | +0.0 | +15.0 |
| | | 1550 | 105560 | -1.1 | -1.5 | +12.2 |
| | | 1640 | 4210 | -1.8 | -7.0 | +2.8 |
| | | 1750 | 110490 | -4.6 | -5.5 | +10.0 |
| | | 1909 | 154670 | -0.3 | -1.3 | +12.4 |
| CI Cyg | SWP5485 | 1550 | - | -1.0 | - | - |
| | | 1909 | - | -0.0 | - | - |

## VIII. References

Bohlin, R.C., Holm, A.V., Savage, B.D., Snijders, M.A.J., and Sparks, W. M. 1980, Astron. and Astrophys., in press.

Bohlin, R. C., and Snijders, M.A.J. 1978, U.S. IUE Newsletter No.2.

Turnrose, B. E., and Harvel, C. 1979, U.S. IUE Newsletter No. 7, p.45.

Turnrose, B. E., Harvel, C., and Bohlin, R.C. 1979, U.S. IUE Newsletter No. 7, p.9.


FIGURE 1
POINT SOURCE
LOW BACKGROUND
100xNET_new/NET_cor
150
140
130
120
110
100
90
80
70
60
50
1000
1100
1200
1300
1400
1500
1600
1700
1800
1900
2000
SCALED NET
WAVELENGTH (Å)


FIGURE 2
POINT SOURCE
LOW BACKGROUND
100xNET new/NET cor
SCALED NET
WAVELENGTH (Å)
150
140
130
120
110
100
90
80
70
60
50
1000
1100
1200
1300
1400
1500
1600
1700
1800
1900
2000


FIGURE 3
EXTENDED SOURCE
MEDIUM BACKGROUND
100xNET new/NET cor
SCALED NET
WAVELENGTH (Å)
150
140
130
120
110
100
90
80
70
60
50
1000
1100
1200
1300
1400
1500
1600
1700
1800
1900
2000


FIGURE 4
POINT SOURCE
HIGH BACKGROUND
100xNET new/NET cor
SCALED NET
WAVELENGTH (Å)
150
140
130
120
110
100
90
80
70
60
50
1000
1100
1200
1300
1400
1500
1600
1700
1800
1900
2000

# APPENDIX

The following FORTRAN coding represents an example of the implementation of the algorithm discussed in the text.

```
      SUBROUTINE SWPFIX(ILINE,NDATA,LSAP,I1,I2,K1,K2,K3,K4,FLUX,WAVE,
     1                  GROSS,BACK)
C
C        PARAMETERS
C          NAME              MEANING
C          ILINE         LINE NUMBER (1-55)
C          NDATA         NO OF POINTS IN THE ARRAYS
C          LSAP          1=LARGE AP.,2=SMALL AP.
C          I1,2          GROSS DEFINITION
C          K1,2,3,4      BACKGROUND DEFINITION (K4<55)
C          FLUX          FNOLD VALUES.IN PLACE CORRECTED (FLUX(I),I=3,NDATA)
C          WAVE          WAVELENTGH                  (WAVE(I),I=3,NDATA)
C          GROSS         GROSS SPECTRUM              (GROSS(I),I=3,NDATA)
C          BACK          BACKGROUND SPECTRUM         (BACK(I),I=3,NDATA)
C
      DIMENSION WL(19), A(19,4,2), B(19,4,2), FR(4,2)
      DIMENSION FLUX(NDATA), WAVE(NDATA), GROSS(NDATA), BACK(NDATA)
      LOGICAL ISG, ISB, ISNOT
      DATA WL/1100.,1150.,1200.,1250.,1300.,1350.,1400.,1450.,1500.,
     1  1550.,1600.,1650.,1700.,1750.,1800.,1850.,1900.,1950., 2000. /
      DATA FR/1084.,2141.,2684.,4291.,1084.,2141.,2688.,4291./
      DATA A/19*0.960,
     1 .2561,.2517,.2725,.2791,.2952,.3103,.3226,.3415,.3415,.3623,
     2 .3557,.3576,.3482,.3415,.3321,.3160,.2573,.2271,.1968 ,1.3674,
     3 1.3830,1.3560,1.3907,1.3771,1.3421,1.3252,1.2900,1.3073,1.2581,
     4 1.2860,1.2580,1.2602,1.2662,1.2729,1.2669,1.3603,1.3945, 1.4132 ,
     5 1.3670,1.3644,1.3602,1.3441,1.3383,1.3399,1.3375,1.3366,1.3309,
     6 1.3333,1.3284,1.3366,1.3424,1.3449,1.3491,1.3627,1.3688,1.3766,
     7  1.3899 ,19*0.960,.2271,
     8 .2431,.2621,.2668,.2971,.3065,.3150,.3169,.3340,.3444,
     9 .3529,.3576,.3557,.3368,.3245,.3236,.2933,.2687,.2365 ,1.4316,
     A 1.4191,1.3906,1.4016,1.3407,1.3175,1.3228,1.3094,1.2739,1.2820,
     B 1.2603,1.2486,1.2407,1.2661,1.2690,1.2616,1.3143,1.3368, 1.3311
     C 1.3636,1.3576,1.3550,1.3485,1.3491,1.3508,1.3433,1.3466,1.3474,
     D 1.3375,1.3391,1.3399,1.3441,1.3483,1.3559,1.3593,1.3610,1.3696,
     E  1.3943 /
      DATA B/19*0.,
     1 763.,768.,746.,738.,721.,704.,691.,671.,671.,648.,
     2 655.,653.,664.,671.,681.,698.,762.,795., 828. ,-1616.,
     3 -1654.,-1574.,-1642.,-1595.,-1504.,-1455.,-1360.,-1397.,-1270.,
     4 -1336.,-1274.,-1269.,-1309.,-1333.,-1337.,-1599.,-1704., -1776.,
     5 -1615.,-1604.,-1585.,-1516.,-1492.,-1499.,-1488.,-1484.,-1460.,
     6 -1470.,-1449.,-1484.,-1509.,-1520.,-1538.,-1596.,-1622.,-1656.,
     7  -1713. ,19*0.,795.,
     8 777.,757.,752.,719.,709.,699.,697.,679.,668.,
     9 658.,653.,655.,676.,689.,690.,723.,750., 785.,-1784.,
     A -1740.,-1659.,-1678.,-1515.,-1456.,-1458.,-1427.,-1333.,-1340.,
     B -1284.,-1254.,-1239.,-1314.,-1333.,-1318.,-1463.,-1537., -1559.,
     C -1600.,-1574.,-1563.,-1534.,-1538.,-1545.,-1513.,-1527.,-1531.,
     D -1488.,-1495.,-1499.,-1516.,-1534.,-1567.,-1582.,-1589.,-1626.,
     E  -1732. /
```

```
      ISG = ILINE.GE.I1.AND.ILINE.LE.I2
      ISB = ILINE.GE.K1.AND.ILINE.LE.K2.OR.
     1     ILINE.GE.K3.AND.ILINE.LE.K4
      IF(ILINE) 1,80,1
 1    IF(ILINE-K4-1) 2,90,99
 2    ISNOT = .NOT.ISG.AND..NOT.ISB
      IF(ISNOT) RETURN
C *** IN PLACE FLUX CORRECTION
      DO 40 IDATA=3,NDATA
         WS = FLUX(IDATA)*.5
         IF(WS-FR(4,LSAP)) 10,10,40
 10      IF(WAVE(IDATA)-1100.) 40,20,20
 20      IF(WAVE(IDATA)-2000.) 30,40,40
 30      J1 = INT(WAVE(IDATA)*.02)-21
         J2 = 1
         IF(WS.GT.FR(1,LSAP).AND.WS.LE.FR(2,LSAP)) J2=2
         IF(WS.GT.FR(2,LSAP).AND.WS.LE.FR(3,LSAP)) J2=3
         IF(WS.GT.FR(3,LSAP).AND.WS.LE.FR(4,LSAP)) J2=4
         ACOEF = (A(J1+1,J2,LSAP)-A(J1,J2,LSAP))*.02
         ACOEF = A(J1,J2,LSAP)+ACOEF*(WAVE(IDATA)-WL(J1))
         BCOEF = (B(J1+1,J2,LSAP)-B(J1,J2,LSAP))*.02
         BCOEF = B(J1,J2,LSAP)+BCOEF*(WAVE(IDATA)-WL(J1))
         FLUX(IDATA) = 2.*(ACOEF*WS+BCOEF)
 40      CONTINUE
      IF(ISB) GO TO 60
C *** GROSS
      DO 50 IDATA = 3,NDATA
         GROSS(IDATA) = GROSS(IDATA)+FLUX(IDATA)
 50      CONTINUE
      RETURN
C *** BACKGROUND
 60   DO 70 IDATA = 3,NDATA
         BACK(IDATA) = BACK(IDATA)+FLUX(IDATA)
 70      CONTINUE
      RETURN
C *** INITIALIZE BACKGROUND AND GROSS
 80   DO 85 IDATA = 3,NDATA
         GROSS(IDATA) = 0.
         BACK(IDATA) = 0.
 85      CONTINUE
      RETURN
C *** NORMALIZE BACKGROUND
 90   SCALE = FLOAT(I2-I1+1)/FLOAT(K2-K1+K4-K3+2)
      DO 95 IDATA = 3,NDATA
         BACK(IDATA) = BACK(IDATA)*SCALE
 95      CONTINUE
 99   RETURN
      END
```